



Atelier « Theremin »


Objectif :

Lors de cet atelier, tu vas réaliser ton propre instrument de musique électronique appelé „Theremin“. Cet instrument est le plus ancien des instruments de musique électronique et c’est aussi le seul instrument dont on joue... sans le toucher ! Ceci se fait en détectant les variations d’intensité lumineuses produites par le mouvement des mains du musicien. La main droite pilote la hauteur de la note, la main gauche quant à elle commande le volume (ON ou OFF). Il s’agit d’une variante de Theremin, l’instrument original utilise deux antennes pour détecter ce mouvement.

Description technique :

Ton Theremin est composé des éléments suivants :


- Un module à microcontrôleur (Arduino Nano)
- Deux phototransistors (pour détecter le mouvement des mains du musicien)
- Trois résistances (1 x 220 Ω et 2 x 47k Ω) et un condensateur 22 μ F 10V
- D’un haut-parleur 8 Ω





Marche à suivre :

1. Place les composants par ordre de hauteur, fais-les vérifier par un animateur avant de les souder.
 - Haut-parleur
 - Résistances
 - Phototransistors **ATTENTION** au sens



- Condensateur **ATTENTION** au sens



- Supports Arduino (**ATTENTION** au sens : voir le connecteur USB). Placer l'Arduino sur les supports avant de souder, cela garantira qu'ils soient perpendiculaires au circuit.

2. Tu dois obtenir ceci :





Programmation :

1. Ouvrir l'application arduino sur l'ordinateur
2. Ouvrir le croquis « theremin »
3. Raccorder le montage à l'ordinateur avec le câble USB
4. Menu Outils :
 - Type de carte : Arduino Nano
 - Processeur : ATmega328P (Old Bootloader)
 - Port : COMx (choisir la plus grande valeur pour x)
5. Cliquer sur le bouton → (Téléverser)

Le montage que tu viens de réaliser permet au microcontrôleur de :

- lire les valeurs mesurées par les phototransistors (en rapport avec la quantité de lumière captée) sur ses broches A0 et A7 (entrées)
- envoyer du son au haut-parleur sur sa broche D8 (sortie)

Il reste maintenant à écrire le programme qui produira le son d'après les valeurs des phototransistors.

Le moyen le plus simple est d'utiliser la fonction `tone()` en lui précisant la broche sur laquelle le son sera émis (D8), la fréquence (hauteur de la note), et éventuellement la durée.

Le but de ce programme est donc de calculer la fréquence de la note d'après la valeur de la lumière captée par le phototransistor A7.

Un programme arduino est constitué de 2 fonctions (suite d'ordres à exécuter) :

- `setup()` : exécutée 1 seule fois au début
- `loop()` : exécutée ensuite, puis rappelée dès qu'elle est terminée. Cette fonction sera toujours exécutée jusqu'à ce que l'on débranche le montage.



Comme les conditions d'éclairage peuvent varier, notre theremin fonctionnera mieux si l'arduino connaît les valeurs correspondantes à l'éclairage minimum (quand on couvre le phototransistor avec nos mains) et à l'éclairage maximum (quand le phototransistor est exposé à la lumière). Cette opération s'appelle le calibrage et sera effectuée dans la fonction `setup()`.

Calibrage : couvrir le phototransistor Q1 après le son grave, le découvrir après le son aigu

Ensuite, il suffit de lire la valeur sur A7 et de faire une règle de 3 pour traduire cette valeur en fréquence.

Exemple : la lumière varie de 1 à 10 et la valeur lue est 3. Comme les fréquences peuvent aller de 100 à 1000, on décidera de jouer la fréquence 300 Hz

Le son ne sera émis que si la valeur lue sur A0 < milieu de l'intervalle de calibrage.

Le programme minimal pourra ressembler à ceci :

```
// variables
int frMin = 440, frMax = 1200; // fréquence minimale et maximale en Hz
int minLight, maxLight;       // valeurs lues au calibrage
int onOff;                     // valeur de déclenchement de la note

// Code exécuté à la mise sous tension ou Reset
void setup() {
  calibrage();
  onOff = (minLight + maxLight) / 2;
}

// Code exécuté en boucle
void loop() {
  // Lecture des phototransistors
  int lectureNote = analogRead(A7); // lecture de la hauteur de la note
  int lectureOnOff = analogRead(A0); // lecture de la commande du volume (ON OFF)
  bool noteOn = (lectureOnOff < onOff);

  // Conversion de la lumière sur A7 en fréquence
  int freq = map(lectureNote, minLight, maxLight, frMin, frMax);

  // Joue la note
  if (noteOn) tone(8, freq, 10); // joue la fréquence freq
}

// Calibrage
void calibrage()
{
  tone(8, 440, 200); // bip qui annonce le début du calibrage
  delay(1200);
  minLight = analogRead(A7); // lecture valeur phototransistor sur A7 recouvert

  tone(8, 880, 200); // bip qui annonce le début du calibrage
  delay(1200);
  maxLight = analogRead(A7); // lecture valeur phototransistor sur A7 découvert

  // si le phototransistor était d'abord découvert, on échange les valeurs
  if (maxLight < minLight) {
    int tmp = minLight;
    minLight = maxLight;
    maxLight = tmp;
  }
}
```

Plus d'infos sur arduino et des exemples de montage : <http://www.arduino.cc>

Contacts : david.grunenwald@he-arc.ch, manon.racine@he-arc.ch